
ABSTRACT Description : This paper explains how to conditionally execute statements in open
code without having to wrap the code in a macro. This is accom-
plished by combining the macro function sysfunc with the data step
function ifc. The result is that parameterized include programs gain
the power of conditional processing — %if ... %then — for which
macros are usually used.

Purpose : replacement for macro if statement

Introduction

Program development proceeds through these stages:Overview

• create ad hoc programs

• identify programs with similar algorithms

• consolidate into parameterized include program

• where necessary, convert to macro for conditional processing

The techniques shown here eliminate the need for macros and replace
conditional processing with inline statements.

Example: %if statement used to conditionally execute blocks of code.

%macro DoThis;
%if <condition> %then %do;

* statements for true;
%end;

%else %do;

* statements for false;
%end;

run;%mend;
%DoThis

can be replaced with this:

%sysfunc(ifc(<condition>
,* statements for true;
,* statements for false;
))

2

Coders' CornerSAS Global Forum 2009

 



Function IFC

The ifc function has four parameters:Data Step Function
IFC 1. a logical expression

2. character value returned when true

3. value returned when false

4. value returned when missing, which is optional
ifc-syntax.txt

1 IFC( logical-expression
2 , value-returned-when-true
3 , value-returned-when-false
4 <, value-returned-when-missing>
5 )

This example shows the function in a data step.
ifc-data.sas

1 DATA Work.TestIfc;
2 attrib N length = 4
3 Value length = $ %length(missing);
4 do N = ., 0, 1, -1;
5 Value = ifc(N
6 ,’true’
7 ,’false’
8 ,’missing’
9 );

10 output;
11 end;
12 stop;
13 run;
14

15 Proc Print data = Work.TestIfc;

ifc-data.lst
1 Obs N Value
2 --- -- -------
3 1 . missing
4 2 0 false
5 3 1 true
6 4 -1 true

Note that minus one is true (or not(false)).

Function SysFunc

The macro function sysfunc provides a method to access certain dataMacro Function
SysFunc step functions and the values of options.

This example shows the combination of sysfunc and getoption writingSysFunc and
GetOption a note to the log with the value of the option source2.

put-sysfunc-getoption.log
1 1 options nosource2;
2 2 %Put option source2 is %sysfunc(getoption(source2));
3 option source2 is NOSOURCE2
4 3 options source2;
5 4 %Put option source2 is %sysfunc(getoption(source2));
6 option source2 is SOURCE2

SysFunc and IFC

This example contains the previous usage of sysfunc and getoption.Macro Statement PUT
with IFC Its purpose is to show that the evaluation of the logical expression can

produce conditional text.
put-sysfunc-ifc.log

1 1 options nosource2;
2 2 %Put option source2 is
3 3 %sysfunc(ifc(%sysfunc(getoption(source2)) eq SOURCE2
4 4 ,True
5 5 ,False
6 6 ,Missing
7 7 ) );
8 option source2 is False

Note that the sysfunc(ifc(...)) is within the %put statement whose
ending semicolon is on log line 7.

The promise of sysfunc(ifc(...)) is to generate statements, basedIFC producing PUT
on the evaluation. This example shows the first test: a complete %put
statement as the result of the evaluation.

The Error
sysfunc-ifc-error.log

1 1 options nosource2;
2 2 %sysfunc(ifc(%sysfunc(getoption(source2)) eq SOURCE2
3 3 ,%Put True;
4 True
5 4 ,%Put False;
6 False
7 5 ,%Put Missing;
8 Missing
9 6 ))

What happened here? All the macro %put statements were executed!

This problem is the same as with call execute, which is explained byNrstr of PUT
Fehd and Carpenter [4], sgf2007.113.The statements must be wrapped
in macro function nrstr (No Rescan String) which delays their execution
until after the ifc condition has been evaluated.

sysfunc-ifc-nrstr.log
1 1 options nosource2;
2 2 %sysfunc(ifc(%sysfunc(getoption(source2)) eq SOURCE2
3 3 ,%nrstr(%Put True;)
4 4 ,%nrstr(%Put False;)
5 5 ,%nrstr(%Put Missing;)
6 6 ) )
7 False

Note that the sysfunc(ifc(...)) is not a statement; the function calls
are completed by the two closing parentheses on log line 6.

Usage Examples

These are common tests:Overview

• boolean: zero, one or greater than one

• comparison of two values

• existence

Many functions return either the set of boolean values — (0,1) — or zeroBoolean
and a negative or positive integer which evaluates as true. Remember that
true is not false, see p. 3.

0,1 : %sysfunc(cexist(&Catalog.))

0, ge 1 : &Nobs.

-1,0, ge 1 : %sysfunc(attrn(&dsid.,nobs))

Any logical expression can be used for the first argument of the ifc func-Comparison
tion.

• &NobsData1. eq &NobsData2.

• &Nobs. gt 100

• %sysfunc(fileref(&FileRef.)) eq 0

• %upcase(%substr(&Type.,1,1)) eq C or &Type. eq 2

There are pitfalls when comparing user-supplied values; refer to Fehd [2],
sas-wiki.Cond-Exec-Global-Stmnts for examples and fixes.

Each of the following programs has been tested as a parameterized includeTest Suite
file, illustrated by this example:

assert-cexist-catalog-Test.sas
1 options source2;
2 Proc Format library = Work; value Test 0=’zero’;run;
3 %Let Catalog = Work.Formats;
4 %Include Project(assert-cexist-catalog);
5 %Let Catalog = Library.Formats;
6 %Include Project(assert-cexist-catalog);

Both the program *.sas and its test *-Test.sas are available in a .zip at
Fehd [2], sas-wiki.Cond-Exec-Global-Stmnts

5

Coders' CornerSAS Global Forum 2009

 

http://www.sascommunity.org/wiki/Conditionally_Executing_Global_Statements
http://www.sascommunity.org/wiki/Conditionally_Executing_Global_Statements


Calling Routines

Evaluation of the logical expression can be used to branch to the executionBranching with
Includes of either of two programs.

cond-inc-which.sas
1 * cond-inc-which;
2 options source2;
3 %Let Data = sashelp.Class;
4 %*Let Data = sashelp.ClassX;* test not exist;
5 %let dsid = %sysfunc(open(&Data. ));
6 %let Nobs =
7 %sysfunc(ifc(&DsId.
8 ,%nrstr( %sysfunc(attrn(&dsid.,nobs));
9 %let rc = %sysfunc(close(&dsid.)); )

10 ,%nrstr(0) ));
11 %sysfunc(ifc(&Nobs.
12 ,%nrstr(%Include Project(cond-inc-1);)
13 ,%nrstr(%Include Project(cond-inc-0);) ))

cond-inc-1.sas
1 %Put Note2: file is cond-inc-1;

cond-inc-0.sas
1 %Put Note2: file is cond-inc-0;

Evaluation of the logical expression can be used to branch to the executionProcedure By Type
of either of two sets of statements.

The elaborate test of the value of Type ensures that this code can be called
with a list processing data set created by either Proc Contents whose Type
are in (1,2) and Proc SQL whose Type are in (’char’,’num’).

proc-type.sas
1 title3 "proc-type of &Libname..&Memname.;";
2 %sysfunc(ifc( %upcase(%substr(&Type.,1,1)) eq C
3 or &Type eq 2
4 ,%nrstr(Proc Freq data = &Libname..&Memname.;
5 tables &Name.;)
6 ,%nrstr(Proc Means data = &Libname..&Memname.;
7 var &Name.;) ))
8 run;

This code is similar to the conditional processing in the SmryEachVar dataCalling Proc-Type
review suite Fehd [3], sgf2008.003.

proc-type-Test.sas
1 options source2;
2 %Let Libname = sashelp;
3 %Let Memname = Class;
4 %Let Name = Sex;
5 %Let Type = C;
6 %Include Project(proc-type);
7

8 %Let Name = Height;
9 %Let Type = 1;

10 %Include Project(proc-type);

proc-type-caller-SQL.sas
1 options source2;
2 %Let Libname = sashelp;
3 %Let Memname = Class;
4 Proc SQL; select catt(’%let Name =’,Name,’;’
5 , ’%let Type =’,Type,’;’
6 ,’%Include Project(proc-type)’)
7 into :List separated by ’;’

8 from Dictionary.Columns
9 where Libname eq "%upcase(&Libname.)"

10 and Memname eq "%upcase(&Memname.)"
11 and Memtype eq ’DATA’;
12 quit;
13 &List.;

proc-type-caller-Contents.sas
1 options source2;
2 %Let Libname = sashelp;
3 %Let Memname = Class;
4 Proc Contents data = &Libname..&Memname.
5 noprint
6 out = Work.Contents;
7 DATA _Null_;
8 attrib Stmnt length = $72;
9 do until(EndoFile);

10 set Work.Contents end = EndoFile;
11 Stmnt = catt(’%Let Name =’,Name,’;’);
12 link ExecStmt;
13 Stmnt = catt(’%Let Type =’,Type,’;’);
14 link ExecStmt;
15 Stmnt = ’%Include Project(proc-type);’;
16 link ExecStmt;
17 end;
18 return;
19 ExecStmt: call execute(catt(’%nrstr(’,Stmnt,’)’));
20 return;
21 stop;
22 run;

The attrn function can be used to get the number of observations of aNumber of Rows
data set. The data set must be opened before the function can return the
value. If the data set exists then the function can be applied; otherwise
when the open of the data set fails, it is not necessary to close the data set.

The values returned by attrn(...,nobs) are in:

-1 : for a view

0 : empty

ge 1 : number of rows
assert-nobs.sas

1 %let dsid = %sysfunc(open(&Data. ));
2 %let Nobs =
3 %sysfunc(ifc(&DsId.
4 ,%nrstr( %sysfunc(attrn(&dsid.,nobs));
5 %let rc = %sysfunc(close(&dsid.)); )
6 ,%nrstr(0) ));
7 %sysfunc(ifc(&Nobs.
8 ,%nrstr(%Put Note2: Data &Data. Nobs: &Nobs.;)
9 ,%nrstr(%Put %sysfunc(sysmsg());

10 endSAS;) ))

As shown above on p. 3 minus one evaluates as true.

21 DATA Work.ClassView / view = Work.ClassView;
22 set sashelp.class(obs=1);
23 run;

NOTE: DATA STEP view saved on file WORK.CLASSVIEW.

24 %Let Data = Work.ClassView;
25 %Include Project(assert-nobs);
Note2: Data Work.ClassView available Nobs: -1

Assertions

Assertions are tests at the beginning of a program that determine whetherConcept
the program input parameters are valid. The assertion of existence follows
the rules of simple boolean evaulation: the functions return zero or one.

Example of Usage

* name: MyProgram;

* confirm input exists;
%let Object = ObjectName;
%Include Asserts(ExistObject);* false==endSAS;

*processing;

The cexist function can be used to test the existence of a catalog.Exist Catalog
assert-cexist-catalog.sas

1 %sysfunc(ifc(%sysfunc(cexist(&Catalog.))
2 ,%nrstr(%Put Note2: Catalog &Catalog. exists;)
3 ,%nrstr(%Put Note3: not exist &Catalog.;
4 endSAS;) ))

The exist function can be used to test whether a data set exists.Exist Data
assert-exist-data.sas

1 %sysfunc(ifc(%sysfunc(exist(&Data.))
2 ,%nrstr(%Put Note2: Data &Data. exists;)
3 ,%nrstr(%Put Note3: not exist &Data.;
4 endSAS;) ))

The fexist function can be used to test the existence of a fileref which isExist Fileref
a directory-specification or folder.

assert-fexist-fileref.sas
1 %sysfunc(ifc(%sysfunc(fexist(&FileRef.))
2 ,%nrstr(%Put Note2: FileRef &FileRef. exists;)
3 ,%nrstr(%Put Note3: not exist &FileRef.;
4 endSAS;) ))

The fileexist function can be used to test the existence of a file-specExist File
ification.

assert-fileexist-filename.sas
1 %sysfunc(ifc(%sysfunc(fileexist(&FileNameExt.))
2 ,%nrstr(%Put Note2: FileNameExt &FileNameExt. exists;)
3 ,%nrstr(%Put Note3: not exist &FileNameExt.;
4 endSAS;) ))

The fileref function is a special case. Instead of returning one as suc-Exist Fileref and File
cess and zero as failure, it returns this set of values:

-1 : fileref exists but file associated with the fileref does not exist

0 : fileref and external file both exist

+1 : fileref not assigned

Thus success is defined as eq 0!

assert-fileref.sas
1 %sysfunc(ifc(%sysfunc(fileref(&FileRef.)) eq 0
2 ,%nrstr(%Put Note2: FileRef &FileRef. exists;)
3 ,%nrstr(%Put Note3: not exist &FileRef.;
4 endSAS;) ))

For an example of usage of function fileref see the OnLine Doc for
function finfo.

Each of these examples show the failure as sufficient to terminate the pro-Summary of Assertion
Usage gram — endSAS; In practice consider adding other programs which pro-

vide information written to a log or data set. Nelson et al. [6], sesug2003.001
discuss a system of event management where subroutines log program
completion and e-mail notifications of failure are sent to interested users.

Conclusion
The combination of functions sysfunc and ifc can be use to eliminate
the need for macros where only conditional processing is needed.

Further Reading

Nelson et al. [6], sesug2003.001 describe assertions in an event manage-Assertions
ment and notification system.

Fehd and Carpenter [4], sgf2007.113 demonstrate the problem of using callCall Execute
and Nrstr execute to generate macro calls and how using function nrstr solves that

problem.

Fehd [1], sugi25.038 provides a trick for conditional execution of includedConditional Execution
of Includes programs.

Henderson [5], saspress.60282 provides a macro to get around the prob-Macro IFF
lem of user-supplied values in comparisons.

Fehd [3], sgf2008.003provides a data review suite of 80 programs withUsing Parameterized
Includes modules, routines and subroutines of parameterized includes.

